

******************************
* Constructing Elections.dta *
******************************

** Presidential Election Data 
use		"$base\Raw Data Files\Election-President.dta"

* generate correlative number to identify timing of elections (useful when there were two elections in one year)
replace		month=6 if year=="1951-2"

replace		year="1951" if year=="1951-2"  | year=="1951-1" 
replace		year="1986" if year=="1986-2" | year=="1986-1" 
replace		year="1992" if year=="1992-2" | year=="1992-1" 
destring year, replace

replace turnout_perc=. if turnout_perc>1
drop pop employment  date 
gen pres=1
save 	"$base\President.dta", replace

** Parliamentary Election Data 
clear
use		"$base\Raw Data Files\Election-Parliament.dta"


* Number of parties running for a seat
egen nparties=rownonmiss(oevp spoe linksblock wdu du ep df dpoe dhug oepu vo pu cdp csp fd boem oenr fpoe kls faoe pv oemp vw bds efpoe dfp lpoe mloe kpoe ndp mrf ol liste grm csa vgoe aloe oep aus gruene gal kaerntner_gruene mir bzoe fritz dc lif rettoe linke_tirol khk linke stark trp matin nfoe ive sau slp cwg demokraten nein oenp dbp bgoe vds wgoe)


* Generate variables of Party votes as a percentage of valid votes
foreach var of varlist oevp-wgoe {
			gen  	`var'_perc=`var'/valid_votes
			replace `var'_perc=0 if `var'==.
}

egen right=rsum(fpoe_perc oevp_perc)
egen left=rsum(spoe_perc kpoe_perc)
gen minor=1-left-right

* Margin of win
egen sh_winner=rowmax(oevp_perc spoe_perc linksblock_perc wdu_perc du_perc ep_perc df_perc dpoe_perc dhug_perc oepu_perc vo_perc pu_perc cdp_perc csp_perc fd_perc boem_perc oenr_perc fpoe_perc kls_perc faoe_perc pv_perc oemp_perc vw_perc bds_perc efpoe_perc dfp_perc lpoe_perc mloe_perc kpoe_perc ndp_perc mrf_perc ol_perc liste_perc grm_perc csa_perc vgoe_perc aloe_perc oep_perc aus_perc gruene_perc gal_perc kaerntner_gruene_perc mir_perc bzoe_perc fritz_perc dc_perc lif_perc rettoe_perc linke_tirol_perc khk_perc linke_perc stark_perc trp_perc matin_perc nfoe_perc ive_perc sau_perc slp_perc cwg_perc demokraten_perc nein_perc oenp_perc dbp_perc bgoe_perc vds_perc wgoe_perc)


foreach var of varlist oevp_perc spoe_perc linksblock_perc wdu_perc du_perc ep_perc df_perc dpoe_perc dhug_perc oepu_perc vo_perc pu_perc cdp_perc csp_perc fd_perc boem_perc oenr_perc fpoe_perc kls_perc faoe_perc pv_perc oemp_perc vw_perc bds_perc efpoe_perc dfp_perc lpoe_perc mloe_perc kpoe_perc ndp_perc mrf_perc ol_perc liste_perc grm_perc csa_perc vgoe_perc aloe_perc oep_perc aus_perc gruene_perc gal_perc kaerntner_gruene_perc mir_perc bzoe_perc fritz_perc dc_perc lif_perc rettoe_perc linke_tirol_perc khk_perc linke_perc stark_perc trp_perc matin_perc nfoe_perc ive_perc sau_perc slp_perc cwg_perc demokraten_perc nein_perc oenp_perc dbp_perc bgoe_perc vds_perc wgoe_perc {
			gen `var'_aux=(`var'-sh_winner)*(-1)
			replace `var'_aux=. if `var'_aux==0
}

egen aux=rowmin(oevp_perc_aux spoe_perc_aux linksblock_perc_aux wdu_perc_aux du_perc_aux ep_perc_aux df_perc_aux dpoe_perc_aux dhug_perc_aux oepu_perc_aux vo_perc_aux pu_perc_aux cdp_perc_aux csp_perc_aux fd_perc_aux boem_perc_aux oenr_perc_aux fpoe_perc_aux kls_perc_aux faoe_perc_aux pv_perc_aux oemp_perc_aux vw_perc_aux bds_perc_aux efpoe_perc_aux dfp_perc_aux lpoe_perc_aux mloe_perc_aux kpoe_perc_aux ndp_perc_aux mrf_perc_aux ol_perc_aux liste_perc_aux grm_perc_aux csa_perc_aux vgoe_perc_aux aloe_perc_aux oep_perc_aux aus_perc_aux gruene_perc_aux gal_perc_aux kaerntner_gruene_perc_aux mir_perc_aux bzoe_perc_aux fritz_perc_aux dc_perc_aux lif_perc_aux rettoe_perc_aux linke_tirol_perc_aux khk_perc_aux linke_perc_aux stark_perc_aux trp_perc_aux matin_perc_aux nfoe_perc_aux ive_perc_aux sau_perc_aux slp_perc_aux cwg_perc_aux demokraten_perc_aux nein_perc_aux oenp_perc_aux dbp_perc_aux bgoe_perc_aux vds_perc_aux wgoe_perc_aux)
gen share_runner=aux*(-1)+sh_winner
gen margin=sh_winner-share_runner

keep  year state_code state num_voters turnout valid_votes invalid_votes valid_total turnout_perc nparties spoe_perc kpoe_perc wdu_perc fpoe_perc oevp_perc  right left minor sh_winner share_runner margin
gen parl=1

save 	"$base\Parliament.dta", replace

** States Election Data 
clear
use		"$base\Raw Data Files\Election-State.dta"

ren	gruene_alternativ_vorarlberg gruene_alt_vorarlberg
ren unabhaengiggen_wahlpartei  wdu

* Number of parties running for a seat
egen		nparties=rownonmiss(fpoe oevp spoe fpoe kpoe wdu linksblock volksopposition demokratische_partei gal_gruene_alternativ vgoe gruene_oesterreichs gruene_altenativ_salzburg gruene liste_tirol trioler_arbeitsbund gruene_alt_vorarlberg gruene_vorarlberg sos_buergerprotestliste lif bl_buergerliste bzoe other_parties)

* Generate variables of Party votes as a percentage of valid votes
foreach var of varlist  oevp spoe fpoe kpoe wdu linksblock volksopposition demokratische_partei gal_gruene_alternativ vgoe gruene_oesterreichs gruene_altenativ_salzburg gruene liste_tirol trioler_arbeitsbund gruene_alt_vorarlberg gruene_vorarlberg sos_buergerprotestliste lif bl_buergerliste bzoe other_parties {
			g 	`var'_perc=`var'/valid_votes
			replace `var'_perc=0 if `var'==.

}

egen		right=rsum(fpoe_perc oevp_perc)
egen		left=rsum(spoe_perc kpoe_perc)
gen minor=1-left-right


* Margin of win
egen		sh_winner=rowmax(fpoe_perc oevp_perc spoe_perc fpoe_perc kpoe_perc wdu_perc linksblock_perc volksopposition_perc demokratische_partei_perc gal_gruene_alternativ_perc vgoe_perc gruene_oesterreichs_perc gruene_altenativ_salzburg_perc gruene_perc liste_tirol_perc trioler_arbeitsbund_perc gruene_alt_vorarlberg_perc gruene_vorarlberg_perc sos_buergerprotestliste_perc lif_perc bl_buergerliste_perc bzoe_perc other_parties_per)

foreach var of varlist fpoe_perc oevp_perc spoe_perc kpoe_perc wdu_perc linksblock_perc volksopposition_perc demokratische_partei_perc gal_gruene_alternativ_perc vgoe_perc gruene_oesterreichs_perc gruene_altenativ_salzburg_perc gruene_perc liste_tirol_perc trioler_arbeitsbund_perc gruene_alt_vorarlberg_perc gruene_vorarlberg_perc sos_buergerprotestliste_perc lif_perc bl_buergerliste_perc bzoe_perc other_parties_perc {
			gen	`var'_x=(`var'-sh_winner)*(-1)
			replace	`var'_x=. if `var'_`x'==0
}

egen aux=rmin(fpoe_perc_x oevp_perc_x spoe_perc_x fpoe_perc_x kpoe_perc_x wdu_perc_x linksblock_perc_x volksopposition_perc_x demokratische_partei_perc_x gal_gruene_alternativ_perc_x vgoe_perc_x gruene_oesterreichs_perc_x gruene_altenativ_salzburg_perc_x gruene_perc_x liste_tirol_perc_x trioler_arbeitsbund_perc_x gruene_alt_vorarlberg_perc_x gruene_vorarlberg_perc_x sos_buergerprotestliste_perc_x lif_perc_x bl_buergerliste_perc_x bzoe_perc_x other_parties_perc_x)
gen share_runner=aux*(-1)+sh_winner
gen margin=sh_winner-share_runner

keep  year state_code state  num_voters turnout valid_votes invalid_votes  valid_total turnout_perc nparties oevp_perc spoe_perc fpoe_perc kpoe_perc wdu_perc right left minor sh_winner share_runner margin

gen st=1

sort	year state_code

** Merge Election Datasets 

append using "$base\Parliament.dta"
append using "$base\President.dta"

replace st=0 if st==.
replace parl=0 if parl==.
replace pres=0 if pres==.

gen invalid= invalid_votes/ turnout
drop  turnout valid_votes invalid_votes valid_total num_voters
rename turnout_perc turnout

*CV dummies
gen CV=0
*Presidential elections
replace CV=1 if pres==1 & year<1986
replace CV=1 if pres==1 & state_code==2 & year<1998
replace CV=1 if pres==1 & state_code==6 & year<1998
replace CV=1 if pres==1 & state_code==7 & year<2010
replace CV=1 if pres==1 & state_code==8 & year<2004
*Parliamentary Elections
replace CV=1 if parl==1 & state_code==2 & (year==1986 | year==1990)
replace CV=1 if parl==1 & state_code==6 & year<1994
replace CV=1 if parl==1 & state_code==7 & year<1994
replace CV=1 if parl==1 & state_code==8 & year<1994
*State Elections
replace CV=1 if st==1 & state_code==2 & year==1989 
replace CV=1 if st==1 & state_code==6 & year<1995
replace CV=1 if st==1 & state_code==7 & year<2003
replace CV=1 if st==1 & state_code==8 & year<2004


* Add Employment Data
merge m:1	year state_code using  "$base\Raw Data Files\population_unemployment_new2.dta"
drop if _merge==2
sort year state_code
gen	l_pop=log(pop_imput)
drop _merge  pop_imput

* Fill in state names
replace state = "Burgenland" if state_code==1
replace state = "Carinthia" if state_code==2
replace state = "Lower Austria" if state_code==3
replace state = "Upper Austria" if state_code==4
replace state = "Salzburg" if state_code==5
replace state = "Styria" if state_code==6
replace state = "Tyrol" if state_code==7
replace state = "Vorarlberg" if state_code==8
replace state = "Vienna" if state_code==9

* Generate election trends 
*Parliamentary 
egen	election_parl=group(year) if parl==1
foreach num of numlist 1(1)9{
gen trend_parl_`num'=0
replace trend_parl_`num'=election_parl if state_code==`num'
replace trend_parl_`num'=0 if  trend_parl_`num'==.
}

drop election_parl 


*Presidential
sort year month
egen	election_pres=group(year month) if pres==1
replace election_pres=0 if pres==0

foreach num of numlist 1(1)9{
gen trend_pres_`num'=0
replace trend_pres_`num'=election_pres if state_code==`num'
replace trend_pres_`num'=0 if  trend_pres_`num'==.

}
drop election_pres 

*State
sort state_code year 
foreach num of numlist 1(1)9{
egen trend_st_`num'=group(year) if st==1 & state_code==`num'
replace trend_st_`num'=0 if trend_st_`num'==.
}

*CV variable for pooled regressions
foreach x in st parl pres{
gen CV_`x'=(CV==1 & `x'==1)
}

drop if year==1945

* CV leads and lags
*Set panel-time
gen election_type=1 if parl==1
replace election_type=2 if st==1
replace election_type=3 if pres==1

sort election_type state_code year month
egen panel=group(state_code election_type)

egen time_parl=group(year) if parl==1
egen time_pres=group(year month) if pres==1

gen time_st=.
foreach x of numlist 1(1)9{
egen time_st_`x'=group(year) if state_code==`x' & st==1
replace time_st=time_st_`x' if state_code==`x'
drop time_st_`x'
}

gen time=time_parl if parl==1
replace time=time_st if st==1
replace time=time_pres if pres==1

tsset panel time

gen lead_CV=F.CV
gen lag_CV=L.CV

replace lag_CV=0 if year==1949 & parl==1 & (state_code<6   | state_code==9)
replace lag_CV=1 if year==1949 & parl==1 & (state_code==6   | state_code==7 | state_code==8 )
replace lag_CV=0 if year==1949 & pres==1
replace lag_CV=0 if year==1949 & st==1 & (state_code<6   | state_code==9)
replace lag_CV=1 if year==1949 & st==1 & (state_code==6   | state_code==7 | state_code==8 )


foreach x in lag lead{
gen CV_parl_`x'=parl*`x'_CV
gen CV_st_`x'=st*`x'_CV
gen  CV_pres_`x'=pres*`x'_CV
}
	
drop  election_type- lag_CV month

tsset, clear


label variable year "Year of election"
label variable state_code "State"
label variable parl "=1 if parliamentary election"
label variable pres "=1 if presidential election"
label variable st "=1 if state election"
label variable turnout "Turnout (%)"
label variable invalid "Invalid Votes (%)
label variable left "Votes for left wing parties (%)"
label variable right "Votes for right wing parties (%)"
label variable minor "Votes for other parties (%)"
label variable nparties "Number of parties"
label variable sh_winner "Vote share of election winner (%)"
label variable margin "Margin of victory (%)"
label variable CV "=1 if voting was compulsory"
label variable CV_parl "CV * Parliamentary Elections"
label variable CV_pres "CV * Presidential Elections"
label variable CV_st "CV * State Elections"
label variable unemployed_perc "Unemployment rate (%)
label variable l_pop "Population (ln)"
label variable  oevp_perc "Vote Share �VP"
label variable  spoe_perc "Vote Share SP�"
label variable  fpoe_perc "Vote Share FP�"
label variable  kpoe_perc "Vote Share KP�"
label variable  wdu_perc "Vote Share VdU"
label variable  CV_parl_lag "CV (t-1) * Parliamentary Elections"
label variable  CV_pres_lag "CV (t-1) * Presidential Elections"
label variable  CV_st_lag "CV (t-1) * State Elections"
label variable  CV_parl_lead "CV (t+1) * Parliamentary Elections"
label variable  CV_pres_lead "CV (t+1) * Presidential Elections"
label variable  CV_st_lead "CV (t+1) * State Elections"
label variable trend_parl_1 "Election specific state-trend - parliamentary elections in state 1"
label variable trend_pres_1 "Election specific state-trend - presidential elections in state 1"
label variable trend_st_1 "Election specific state-trend - state elections in state 1"
label variable trend_parl_2 "Election specific state-trend - parliamentary elections in state 2"
label variable trend_pres_2 "Election specific state-trend - presidential elections in state 2"
label variable trend_st_2 "Election specific state-trend - state elections in state 2"
label variable trend_parl_3 "Election specific state-trend - parliamentary elections in state 3"
label variable trend_pres_3 "Election specific state-trend - presidential elections in state 3"
label variable trend_st_3 "Election specific state-trend - state elections in state 3"
label variable trend_parl_4 "Election specific state-trend - parliamentary elections in state 4"
label variable trend_pres_4 "Election specific state-trend - presidential elections in state 4"
label variable trend_st_4 "Election specific state-trend - state elections in state 4"
label variable trend_parl_5 "Election specific state-trend - parliamentary elections in state 5"
label variable trend_pres_5 "Election specific state-trend - presidential elections in state 5"
label variable trend_st_5 "Election specific state-trend - state elections in state 5"
label variable trend_parl_6 "Election specific state-trend - parliamentary elections in state 6"
label variable trend_pres_6 "Election specific state-trend - presidential elections in state 6"
label variable trend_st_6 "Election specific state-trend - state elections in state 6"
label variable trend_parl_7 "Election specific state-trend - parliamentary elections in state 7"
label variable trend_pres_7 "Election specific state-trend - presidential elections in state 7"
label variable trend_st_7 "Election specific state-trend - state elections in state 7"
label variable trend_parl_8 "Election specific state-trend - parliamentary elections in state 8"
label variable trend_pres_8 "Election specific state-trend - presidential elections in state 8"
label variable trend_st_8 "Election specific state-trend - state elections in state 8"
label variable trend_parl_9 "Election specific state-trend - parliamentary elections in state 9"
label variable trend_pres_9 "Election specific state-trend - presidential elections in state 9"
label variable trend_st_9 "Election specific state-trend - state elections in state 9"
label define state_lbl 1 "Burgenland" 2 "Carinthia" 3 "Lower Austria" 4 "Upper Austria" 5 "Salzburg" 6 "Styria" 7 "Tyrol" 8 "Vorarlberg" 9 "Vienna"
label values state_code state_lbl
drop state share_runner

save "$base\Elections.dta", replace


*********************************
* Constructing Expenditures.dta *
*********************************
clear
use		"$base\Raw Data Files\1980-2012_expenditures.dta"
sort	year state_code

merge m:1 year using "$base\Raw Data Files\cpi_index.dta"
drop if _merge==2
drop _merge

*Express it all in real terms
foreach x in admin security educ culture welfare health transport economy services finance{
replace `x'_exp=`x'_exp*100/cpi
}


*Add information on population
merge 1:1	year state_code using  "$base\Raw Data Files\population_unemployment_new2.dta"
drop if _merge==2
drop   unemployed_perc _merge cpi

gen total_exp=  admin_exp+ security_exp+ educ_exp+ culture_exp+ welfare_exp+ health_exp+ transport_exp+ economy_exp+ services_exp+ finance_exp

rename admin_exp gen_exp
rename welfare_exp social_exp
rename culture_exp arts_exp

* Generate per capita variables
g total_exp_pc=total_exp/pop_imput
foreach var of varlist gen_exp- finance_exp {
			g `var'_pc=`var'/pop_imput
}

*We shift expenditures one year back (because an election determines expenditures of the next year, not of the current one)
replace year=year-1


* Aggregate Expenditure Categories 
gen l_total_exp=ln(total_exp)
gen l_total_exp_pc=ln(total_exp_pc)

g admin_exp=gen_exp + finance_exp
g admin_exp_perc=(admin_exp)/total_exp
g l_admin_exp=ln(admin_exp)
g l_admin_exp_pc=ln(admin_exp/pop_imput)

g welfare_exp=(educ_exp + social_exp+ health_exp+ arts_exp)
g welfare_exp_perc=(welfare_exp)/total_exp
g l_welfare_exp=ln(welfare_exp)
g l_welfare_exp_pc=ln(welfare_exp/pop_imput)

g infras_exp=(transport_exp + security_exp +economy_exp+services_exp)
g infras_exp_perc=(infras_exp)/total_exp 
g l_infras_exp=ln(infras_exp) 
g l_infras_exp_pc=ln(infras_exp/pop_imput)

drop pop_imput

*Generate these log measures for all the subcategories as well
foreach x in gen finance educ social health arts transport security economy services{
gen l_`x'_exp=ln(`x'_exp)
gen l_`x'_exp_pc=ln(`x'_exp_pc)
gen `x'_exp_perc=`x'_exp/total_exp
}

 drop   total_exp_pc- finance_exp_pc 



*Add unemployment and population data
merge 1:1	year state_code using  "$base\Raw Data Files\population_unemployment_new2.dta"
keep if _merge==3
gen l_pop=log(pop_imput)
drop _merge
gen pop_thousands=pop_imput/1000
drop pop_imput

* Generate state-year trends for the expenditure regressions 
egen trend=group(year)
foreach num of numlist 1(1)9{
gen trend_`num'=0
replace trend_`num'=trend if state_code==`num'
}
drop trend


save 	"$base\Expenditures.dta", replace


*Add the turnout data
clear
use "$base\Elections" 
*Use the turnout data from the 1st round if there were runoffs in presidential elections
drop if pres==1 & (trend_pres_1==10 | trend_pres_2==10 | trend_pres_3==10  | trend_pres_4==10  | trend_pres_5==10  | trend_pres_6==10  | trend_pres_7==10  | trend_pres_8==10  | trend_pres_9==10)
drop if pres==1 & (trend_pres_1==12 | trend_pres_2==12 | trend_pres_3==12  | trend_pres_4==12 | trend_pres_5==12  | trend_pres_6==12  | trend_pres_7==12  | trend_pres_8==12  | trend_pres_9==12)

keep if year>1960

gen election_type="parl" if parl==1
replace election_type="st" if st==1
replace election_type="pres" if pres==1

gen CV_lead=CV_parl_lead if parl==1
replace CV_lead=CV_st_lead if st==1
replace CV_lead=CV_pres_lead if pres==1
gen CV_lag=CV_parl_lag if parl==1
replace CV_lag=CV_st_lag if st==1
replace CV_lag=CV_pres_lag if pres==1


keep year state_code  CV CV_lead CV_lag turnout election_type
rename turnout turnout_
rename CV CV_
rename CV_lead CV_lead_
rename CV_lag CV_lag_
reshape wide turnout_ CV_ CV_lead_ CV_lag_, i(year state_code) j(election_type) string

cd "$base"
merge 1:1 year state_code using "$base\Expenditures.dta"


** Note: we have turnout>1 in the 1974 pres elections for states 1 and 3, and in the 1980 presidential elections for state 1
gen no_carryforward=1 if state_code==1 & (year>1973 & year<1986)
replace no_carryforward=1 if state_code==3 & (year>1973 & year<1980)


* Replace data from missing years with data from previous years (install the carryforward package)
sort state_code year
foreach var of varlist  turnout_parl CV_parl CV_lead_parl CV_lag_parl turnout_pres CV_pres CV_lead_pres CV_lag_pres turnout_st CV_st CV_lead_st CV_lag_st{
	bysort state_code: carryforward `var', replace
	}

replace turnout_pres=. if no_carryforward==1

foreach x in lag lead{
foreach y in parl pres st{
rename CV_`x'_`y' CV_`y'_`x'
}
}

*Correct the leads for presidential elections when there's a runoff (in 1992, the following election (1998) had no CV for Carinthia, for example).
replace CV_pres_lead=0 if year>1991 & year<1998 & (state_code==2 | state_code==6)
keep if gen_exp!=.
drop  pop_thousands
drop  state  _merge no_carryforward 
label variable year "Year of election"
label variable state_code "State"
label variable turnout_parl "Turnout in previous parliamentary election (%)"
label variable turnout_pres "Turnout in previous presidential election (%)"
label variable turnout_st "Turnout in previous state election (%)"
label variable CV_parl "CV in electoral period in Parliamentary Elections"
label variable CV_pres "CV in electoral period in Presidential Elections"
label variable CV_st "CV in electoral period in State Elections"
label variable  CV_parl_lag "CV in past electoral period in Parliamentary Elections"
label variable  CV_pres_lag "CV in past electoral period in Presidential Elections"
label variable  CV_st_lag "CV in past electoral period in State Elections"
label variable  CV_parl_lead "CV in next electoral period in Parliamentary Elections"
label variable  CV_pres_lead "CV in next electoral period in Presidential Elections"
label variable  CV_st_lead "CV in next electoral period in State Elections"
label variable trend_1 "Yearly specific state-trend - state 1"
label variable trend_2 "Yearly specific state-trend - state 2"
label variable trend_3 "Yearly specific state-trend - state 3"
label variable trend_4 "Yearly specific state-trend - state 4"
label variable trend_5 "Yearly specific state-trend - state 5"
label variable trend_6 "Yearly specific state-trend - state 6"
label variable trend_7 "Yearly specific state-trend - state 7"
label variable trend_8 "Yearly specific state-trend - state 8"
label variable trend_9 "Yearly specific state-trend - state 9"
label variable unemployed_perc "Unemployment rate (%)
label variable l_pop "Population (ln)"
label variable  total_exp "Expenditures - Total"
label variable  l_total_exp "Expenditures (in Ln) - Total"
label variable  l_total_exp_pc "Expenditures (in Ln of per capita) - Total"
label variable  admin_exp "Expenditures - Administrative"
label variable  admin_exp_perc "Expenditures (as % of Total) - Administrative"
label variable  l_admin_exp "Expenditures (in Ln) - Administrative"
label variable  l_admin_exp_pc "Expenditures (in Ln of per capita) - Administrative"
label variable  welfare_exp "Expenditures - Welfare"
label variable  welfare_exp_perc "Expenditures (as % of Total) - Welfare"
label variable  l_welfare_exp "Expenditures (in Ln) - Welfare"
label variable  l_welfare_exp_pc "Expenditures (in Ln of per capita) - Welfare"
label variable  infras_exp "Expenditures - Infrastructure"
label variable  infras_exp_perc "Expenditures (as % of Total) - Infrastructure"
label variable  l_infras_exp "Expenditures (in Ln) - Infrastructure"
label variable  l_infras_exp_pc "Expenditures (in Ln of per capita) - Infrastructure"
label variable  gen_exp "Expenditures - Representatives & Gen. Admin."
label variable  gen_exp_perc "Expenditures (as % of Total) - Representatives & Gen. Admin."
label variable  l_gen_exp "Expenditures (in Ln) - Representatives & Gen. Admin."
label variable  l_gen_exp_pc "Expenditures (in Ln of per capita) - Representatives & Gen. Admin."
label variable  finance_exp "Expenditures - Finance"
label variable  finance_exp_perc "Expenditures (as % of Total) - Finance"
label variable  l_finance_exp "Expenditures (in Ln) - Finance"
label variable  l_finance_exp_pc "Expenditures (in Ln of per capita) - Finance"
label variable  educ_exp "Expenditures - Education"
label variable  educ_exp_perc "Expenditures (as % of Total) - Education"
label variable  l_educ_exp "Expenditures (in Ln) - Education"
label variable  l_educ_exp_pc "Expenditures (in Ln of per capita) - Education"
label variable  social_exp "Expenditures - Social"
label variable  social_exp_perc "Expenditures (as % of Total) - Social"
label variable  l_social_exp "Expenditures (in Ln) - Social"
label variable  l_social_exp_pc "Expenditures (in Ln of per capita) - Social"
label variable  health_exp "Expenditures - Health"
label variable  health_exp_perc "Expenditures (as % of Total) - Health"
label variable  l_health_exp "Expenditures (in Ln) - Health"
label variable  l_health_exp_pc "Expenditures (in Ln of per capita) - Health"
label variable  arts_exp "Expenditures - Arts"
label variable  arts_exp_perc "Expenditures (as % of Total) - Arts"
label variable  l_arts_exp "Expenditures (in Ln) - Arts"
label variable  l_arts_exp_pc "Expenditures (in Ln of per capita) - Arts"
label variable  transport_exp "Expenditures - Transport"
label variable  transport_exp_perc "Expenditures (as % of Total) - Transport"
label variable  l_transport_exp "Expenditures (in Ln) - Transport"
label variable  l_transport_exp_pc "Expenditures (in Ln of per capita) - Transport"
label variable  security_exp "Expenditures - Security"
label variable  security_exp_perc "Expenditures (as % of Total) - Security"
label variable  l_security_exp "Expenditures (in Ln) - Security"
label variable  l_security_exp_pc "Expenditures (in Ln of per capita) - Security"
label variable  economy_exp "Expenditures - Prom. of the economy"
label variable  economy_exp_perc "Expenditures (as % of Total) - Prom. of the economy"
label variable  l_economy_exp "Expenditures (in Ln) - Prom. of the economy"
label variable  l_economy_exp_pc "Expenditures (in Ln of per capita) - Prom. of the economy"
label variable  services_exp "Expenditures - Services"
label variable  services_exp_perc "Expenditures (as % of Total) - Services"
label variable  l_services_exp "Expenditures (in Ln) - Services"
label variable  l_services_exp_pc "Expenditures (in Ln of per capita) - Services"


save 	"$base\Expenditures.dta", replace










